Reprinted from: © Journal of the American Statistical Association
June 1976, Volume 71, Number 354
Theory and Methods Section
Pages 340–344

A Method for Simulating Stable Random Variables

J. M. CHAMBERS, C. L. MALLOWS and B. W. STUCK

A Method for Simulating Stable Random Variables

J. M. CHAMBERS, C. L. MALLOWS and B. W. STUCK*

A new algorithm is presented for simulating stable random variables on a digital computer for arbitrary characteristic exponent $\alpha (0 < \alpha \leq 2)$ and skewness parameter $\beta (-1 \leq \beta \leq 1)$. The algorithm involves a nonlinear transformation of two independent uniform random variables into one stable random variable. This stable random variable is a continuous function of each of the uniform random variables, and of α and a modified skewness parameter β' throughout their respective permissible ranges.

1. INTRODUCTION

In recent years there has been growing interest in the properties and uses of stable distributions, which are those distributions that arise as limits of suitably scaled and translated sums of iid random variables. An excellent reference for this theory is Feller [9, e.g., pp. 165–73; 540–9]. This note presents a new solution to the problem of how to simulate stable random variables on a digital computer.

To fix notation, we take the characteristic function $f(t; \alpha, \beta)$ (with a location parameter of zero and a scale or dispersion parameter of unity) to be

$$f(t; \alpha, \beta) = \exp\left[-|t|^{\alpha} \exp\left(-\tfrac{1}{2}\pi i \beta k(\alpha) \operatorname{sign}(t)\right)\right], \quad 0 < \alpha \leq 2, \quad \alpha \neq 1,$$

$$= \exp\left[-|t|\left(1 + \frac{2}{\pi} i \beta \ln |t| \operatorname{sign}(t)\right)\right], \quad \alpha = 1,$$

where $-1 \leq \beta \leq 1$, $k(\alpha) = 1 - |1 - \alpha|$. For $\alpha \neq 1$ this is a slight modification of [9, p. 548, Eqs. (6.1) and (6.2)]. Feller [9, p. 542, Eq. (4.10)] incorrectly omits the factor $(2/\pi)$ for $\alpha = 1$. (The reader should also consult Zolotarev [19].)

Previously, stable random variables have been simulated by generating independent pseudorandom numbers distributed uniformly on the interval (0, 1) and calculating the (approximate) inverse distribution function at each value, the result being a sequence of iid stable pseudorandom variables (see [5, 8, 15]). DuMouchel [6, pp. 121–4] has considered a number of variations on this method. Such methods are reasonable when the number of simulations needed is large, for a single (α, β) pair, since it is then practicable to put much effort into the preliminary calculation of the inverse distribution function. However, this step is expensive, and any inaccuracies introduced will be transmitted to the Monte Carlo results. For problems of small or moderate size, especially when many different (α, β) parameter values are of interest, the algorithm developed here will be more efficient and convenient.

In several cases, special methods can be used. Three of these are the Gaussian, $S(2, 0)$, the Cauchy, $S(1, 0)$ and the Pearson V, $S(\frac{1}{2}, 1)$. Our method reduces to well-known formulas in these cases. A procedure for a set of special cases is given by Brown and Tukey [3]: they take m iid Gaussian random variables $(X_1, \ldots, X_m)$, with zero mean and unit variance, and form $S(2^{-m}, 1)$ from

$$S(2^{-m}, 1) = \prod_{j=1}^{m} (X_j)^{-2^j}$$

Further cases can be generated from the preceding by using the following relationships, all of which follow immediately from the manipulation of the characteristic function $f(t; \alpha, \beta)$ plus elementary Fourier transform properties (we use the notation "$\sim$" to mean "has the same distribution as"):

i. $S(\alpha, \beta) \sim -S(\alpha, -\beta)$

ii. $S(\alpha, \beta) \sim pS_1 - qS_2$,

where S_1, S_2 are iid, $S_1 \sim S_2 \sim S(\alpha, 1)$, and

$$p^{\alpha} = \sin\left(\tfrac{1}{2}\pi k(\alpha)(1 + \beta)\right)/\sin\left(\pi k(\alpha)\right)$$

$$q^{\alpha} = \sin\left(\tfrac{1}{2}\pi k(\alpha)(1 - \beta)\right)/\sin\left(\pi k(\alpha)\right)$$

iii. $S(\alpha'', \beta') \sim S(\alpha, \beta) S(\alpha', 1)^{1/\alpha}$, $\alpha \neq 1, \alpha' < 1$,

where $\alpha'' = \alpha\alpha'$, $\beta' = \beta k(\alpha)\alpha'/k(\alpha\alpha')$. (See also [9, p. 336].)

The method presented here is an extension of an observation of Kanter [13, 14]. The manuscript [13] contains a recipe for simulating positive ($\beta = 1$) stable random variables ($0 < \alpha < 1$); it appears at (2.2). This was found by interpreting an integral representation given (slightly incorrectly) by Ibragimov and Chernin [12] for a positive stable probability density function. A single stable random variable can then be generated with a predetermined skewness via (ii) in the preceding.

* J.M. Chambers, C.L. Mallows and B.W. Stuck are members of technical staff, and C.L. Mallows is department head, all at Bell Laboratories, Murray Hill, NJ 07974.

© Journal of the American Statistical Association
June 1976, Volume 71, Number 354
Theory and Methods Section

Here we give similar recipes for all stable random variables of arbitrary index $\alpha (0 < \alpha \leq 2)$ and skewness parameter $\beta (-1 \leq \beta \leq 1)$, (2.3) and (2.4). These results follow immediately from representations given by Zolotarev [19] for the probability density functions of the general stable laws. The unnecessary restriction of generating a single stable random variable from a linear combination of two maximally skew ($\beta = 1$) stable random variables is dropped; this suggests our algorithm may be computationally more efficient (see Section 5).

A difficulty in (2.3) and (2.4) is that $S(\alpha, \beta)$ is not a continuous function of α in the neighborhood of $\alpha = 1$. We discuss this issue in Section 3, and reparameterize $S(\alpha, \beta)$ to overcome this problem. In Section 4 the numerical implementation of our algorithm is discussed, with care taken to insure high relative accuracy. Section 5 touches briefly on the issues of accuracy, speed and convenience.

This work may find application in gaining insight via Monte Carlo methods into problems associated with such seemingly diverse areas as communications engineering and finance.

Recent work [18] has indicated that noise on some telephone lines may be adequately modeled as the sum of a deterministic process (e.g., sinusoids) and a non-Gaussian stable random process. Eventually this model may be used to design equipment for detecting and estimating signals corrupted by this noise. Some indication of the performance of this equipment can be obtained by computer simulation, using the method presented here for generating stable variates.

Over the past decade, a growing body of evidence [4, 7, 17] has accumulated indicating stock prices can be adequately modeled as the sum (or product) of independent identically distributed (iid) random variables, with the random variables (or their logarithms, respectively) drawn from a non-Gaussian stable distribution. It would be desirable to use this model to formulate intelligent investment strategies. Again, some indication of the performance of a particular strategy can be gained through computer simulation using the method presented here.

In Monte Carlo studies of various location and scale parameter estimates [1, 10, 16] the use of representations of the form $S = NV$ where N is Normal, independent of V, have led to substantial increases in efficiency. Such studies can now easily encompass the symmetric stable family, using the algorithm presented here.

2. THE NEW FORMULAS

For the case $0 < \alpha < 1$, $\beta = 1$, Ibragimov and Chernin [12] derived a formula for the density of $S(\alpha, 1)$; this can be integrated to yield, for $x > 0$,

$$P[S(\alpha, 1) \leq x] = \frac{1}{\pi} \int_0^{\pi} \exp\left[-x^{\alpha/(1-\alpha)} a(\theta)\right] d\theta, \quad 0 < \alpha < 1, \quad (2.1)$$

where

$$a(\theta) = \frac{\sin\left((1-\alpha)\theta\right)(\sin \alpha\theta)^{\alpha/(1-\alpha)}}{(\sin \theta)^{1/(1-\alpha)}}, \quad 0 < \theta < \pi.$$

Noticing that for a standard exponential variable W we have $P[W \geq w] = e^{-w} (w \geq 0)$, Kanter [13] deduced from (2.1) the representation

$$S(\alpha, 1) = \left(a(\Theta)/W\right)^{(1-\alpha)/\alpha}, \quad (2.2)$$

where Θ is uniform on $(0, \pi)$ and W is standard exponential, and Θ and W are mutually independent.

Zolotarev [19] gave expressions similar to (2.1) in the general case; hence, we can deduce the representations

$$S(\alpha, \beta) = \frac{\sin \alpha(\Phi - \Phi_0)}{(\cos \Phi)^{1/\alpha}} \cdot \left(\frac{\cos\left(\Phi - \alpha(\Phi - \Phi_0)\right)}{W}\right)^{(1-\alpha)/\alpha}, \quad \alpha \neq 1$$

$$S(1, \beta) = \frac{2}{\pi}\left(\left(\tfrac{1}{2}\pi + \beta\Phi\right) \tan \Phi - \beta \ln\left(\frac{\frac{1}{2}\pi W \cos \Phi}{\frac{1}{2}\pi + \beta\Phi}\right)\right), \quad (2.4)$$

where W is as before and Φ is uniform on $(-\frac{1}{2}\pi, \frac{1}{2}\pi)$; also $\Phi_0 = -\frac{1}{2}\pi\beta(k(\alpha)/\alpha)$. Notice that (2.3) reduces to (2.2) when $\beta \to 1 (\Phi_0 \to -\frac{1}{2}\pi)$, $\alpha < 1$ ($\Theta = \frac{1}{2}\pi + \Phi$).

Certain special cases of (2.3) and (2.4) are interesting. With $\alpha = 2$, $\beta = 0$, we have

$$S(2, 0) = W^{\frac{1}{2}} \sin 2\Phi / \cos \Phi = 2W^{\frac{1}{2}} \sin \Phi,$$

which is the Box-Muller representation of a Normal random variable [2]. With $\alpha = 1$, $\beta = 0$, we find for the Cauchy case,

$$S(1, 0) = \tan \Phi.$$

With $\alpha = \frac{1}{2}$, $\beta = 1 (\Phi_0 = -\frac{1}{2}\pi)$, we have

$$S(\tfrac{1}{2}, 1) = (4W \sin^2(\tfrac{1}{2}(\Phi - \tfrac{1}{2}\pi)))^{-1} \sim (4W \cos^2 \Theta)^{-1}.$$

This demonstrates the well-known fact that $S(\frac{1}{2}, 1)$ is distributed like $1/S^2(2, 0)$.

3. TRANSITIONAL BEHAVIOR IN THE NEIGHBORHOOD OF $\alpha = 1$

In this section we study the behavior of (2.3) as $\alpha \to 1$. In related work, DuMouchel [6, pp. 13–14] has studied this problem for a Cartesian parameterization of the stable characteristic function rather than the polar parameterization we have adopted.

With β fixed and positive, if we let $\alpha \to 1$ in (2.3), then we obtain

$$\lim_{\alpha \to 1} S(\alpha, \beta) = \frac{\sin(\Phi - \Phi_0)}{\cos(\Phi)} = \cos \varphi_0 \tan \Phi - \sin \varphi_0,$$

which is simply a linear function of a Cauchy ($\alpha = 1$, $\beta = 0$) random variable. If now we let $\beta \to 1$, this limit becomes degenerate (at $+1$). To obtain (2.4) as a limit,

we claim that we must translate and rescale $S(\alpha, \beta)$, and must also let $\beta \to 1$ and $\alpha \to 1$.

To see this, it is convenient to define

$$S'(\alpha, \beta) = \tan \alpha\Phi_0 + (\cos \alpha\Phi_0)^{-1/\alpha} S(\alpha, \beta) , \quad \alpha \neq 1 , \quad (3.1)$$

which has characteristic function f' given by

$$f'(t; \alpha, \beta) = \exp \{-|t|^\alpha + it(1 - |t|^{\alpha-1}) \tan (\alpha\Phi_0)\} .$$

We define $S'(1, \beta) = S(1, \beta)$, $S'(2, 0) = S(2, 0)$. The limiting behavior of $S'(\alpha, \beta)$ as $\alpha \to 1$ can be deduced from its characteristic function. To obtain the standard form (2.4) at $\alpha = 1$, β must depend on α as follows.

Theorem: Let $\epsilon = 1 - \alpha$, $\sigma = \pm 1, \lambda > 0$ be given. Take $\beta = \sigma(1 - \lambda|\epsilon|)$. Then

i. as $\epsilon \downarrow 0$, $S'(\alpha, \beta) \to S'(1, \beta^*)$ with $\beta^* = \sigma/(1 + \lambda)$

ii. as $\epsilon \uparrow 0$, $S'(\alpha, \beta) \to S'(1, \beta^*)$ with $\beta^* = -\sigma/(1 + \lambda)$.

The proof is elementary and is omitted.

This discontinuity in limiting behavior is awkward in applications that extend over an interval of values of α in the neighborhood of $\alpha = 1$. For such purposes, and to retain computational accuracy, it is convenient to reparameterize the family of stable distributions, replacing β by β', where

$$\beta'(\alpha, \beta) = -\tan\left(\tfrac{1}{2}\pi(1 - \alpha)\right) \tan (\alpha\Phi_0) , \quad \alpha \neq 1 ,$$
$$\beta'(1, \beta) = \beta , \quad \alpha = 1 . \quad (3.2)$$

(Recall Φ_0 was defined after (2.4)). In this notation, the characteristic function of $S'(\alpha, \beta')$ is $f'(t; \alpha, \beta')$,

$$f'(t; \alpha, \beta') = \exp \{-|t|^\alpha - it(1 - |t|^{\alpha-1})\beta' \tan\left(\tfrac{1}{2}\pi\alpha\right)\} , \quad \alpha \neq 1 , \quad (3.3)$$
$$f'(t; 1, \beta) = f(t; 1, \beta) , \quad \alpha = 1 .$$

Both $f'(t; \alpha, \beta')$ and $S'(\alpha, \beta)$ are continuous functions of α in the neighborhood of $\alpha = 1$. Thus, this new parameterization of the family of stable distributions is continuous in both α and β', as claimed.

4. COMPUTATION

The computation of $S(\alpha, \beta)$ from (2.3), (2.4) is straightforward, except in the limiting case as $\alpha \to 1$ and in the degenerate case $\alpha \downarrow 0$. In this section, we discuss one approach for computing $S'(\alpha, \beta')$ using the revised parameterization (α, β') of (3.2). Again, we emphasize that this permits accurate representation near $\alpha = 1$, and that S' is a continuous function of all its arguments.

We begin by noting that $S'(\alpha, \beta')$ can be written as

$$S' = \tan \alpha\Phi_0 + \frac{\sin \alpha\Phi - \tan \alpha\Phi_0 \cos \alpha\Phi}{\cos \Phi} z^{\epsilon/(1-\epsilon)} , \quad (4.1)$$

where $\epsilon = 1 - \alpha$,

$$z = [(\cos \epsilon\Phi - \tan \alpha\Phi_0 \sin \epsilon\Phi)/(W \cos \Phi)] .$$

Because $\epsilon\Phi \to 0$ and $\tan \alpha\Phi_0 \to \infty$ as $\epsilon \to 0$, it is necessary to rearrange the calculations. At the same time, one can reduce the trigonometric computations to the tangent function of half angles. To retain accuracy throughout the range, the algorithm uses two auxiliary functions,

$$D_2(x) = (\exp(x) - 1)/x$$

and

$$\tan_2(x) = \tan(x)/x . \quad (4.2)$$

Although these are nonstandard functions, approximations to them may be developed from existing approximations to $D(x) = \exp(x) - 1$ and $\tan(x)$. The latter functions are often computed by functions of the form $xR(x)$, where $R(x)$ is a rational function of special form; thus, we can use $R(x)$ to approximate the functions in (4.2). (Note, however, that standard range reduction may not carry over.) In the appendix, FORTRAN approximations to $D_2(x)$ and $\tan_2(x)$ are given, based on approximations 1801 and 4283, respectively, of [11]. These are adequate for precision of about 27 bits. Readers may adapt more or less precise routines from the same source for their own needs. In our algorithm, the quantities shown in the table are explicitly evaluated.

Expressions for Computation

Symbol	Computation		
	$\begin{cases} \beta'/(\frac{1}{2}\pi \tan_2(\frac{1}{2}\pi\epsilon)) \\ \frac{1}{2}\pi\beta'\epsilon(1-\epsilon) \tan_2(\frac{1}{2}\pi(1-\epsilon)) \end{cases}$	$\epsilon > -.99$ $\epsilon \leq -.99$	$-\epsilon \tan \alpha\Phi_0$
	$\frac{1}{2}\Phi \tan_2(\frac{1}{2}\Phi)$		$\tan(\frac{1}{2}\Phi)$
	$\tan_2(\frac{1}{2}\epsilon\Phi)$		$\tan(\frac{1}{2}\epsilon\Phi)/(\frac{1}{2}\epsilon\Phi)$
	$\frac{1}{2}\epsilon\Phi B$		$\tan(\frac{1}{2}\epsilon\Phi)$
	$\frac{(1+a^2)(1-b^2+\tau B B)}{W(1-a^2)(1+b^2)}$		$\frac{\cos \epsilon\Phi - \tan \alpha\Phi_0 \sin \epsilon\Phi}{W \cos \Phi}$
d	$\log(z) \frac{D_2(\epsilon \log z/(1-\epsilon))}{1-\epsilon}$		$\frac{z^{\epsilon/1-\epsilon} - 1}{\epsilon}$

An accurately computable rearrangement of (4.1) is

$$S' = \left(\frac{\sin \alpha\Phi}{\cos \Phi} - \tan \alpha\Phi_0 \left(\frac{\cos \alpha\Phi}{\cos \Phi} - 1\right)\right) z^{\epsilon/(1-\epsilon)} + \tan \alpha\Phi_0 (1 - z^{\epsilon/(1-\epsilon)}) ,$$

which, in terms of the symbols in the table, is evaluated as

$$S' = \left(\frac{2(a - b)(1 + ab) - \Phi\tau B(b(1 - a^2) - 2a)}{((1 - a^2)(1 + b^2))}\right) \cdot (1 + \epsilon d) + \tau d .$$

The FORTRAN function RSTAB in the Appendix implements this calculation.

5. COMPARISONS: ACCURACY, SPEED AND CONVENIENCE

The algorithm described here provides a very convenient general method of simulating random variables from any stable distribution, and for most purposes is believed to be equal or superior to all known existing procedures in accuracy and speed. For many applications,

the generality and ease of use may well be the most important criteria. The FORTRAN function RSTAB may be used with any high-quality generators of uniform and exponential pseudorandom numbers. (If a special exponential generator is not available, an exponential random variable may be generated as minus the natural logarithm of a second uniform.)

Two alternative methods are the special case (2.2) of a most asymmetric stable distribution, and the use of numerical inversion of the characteristic function as in the work of DuMouchel [6, pp. 35–42]. As noted in Section 1, the former applies only for $\alpha < 1$. For this range, an arbitrary $S(\alpha, \beta)$ can be computed by a suitable linear combination of two $S(\alpha, 1)$ values. However, the implementation of (2.2) appears to require nearly as much computation as the more general case; in particular, an accurately computable form of (2.2), obtained by specializing the formula for S' in Section 4, still requires all the special functions calls (three to "tan$_2$" two to "log", and one to "D_2"). Thus, the direct evaluation of the general case should be nearly twice as fast. In addition, the linear combination required may be numerically inaccurate for a skewness parameter close to zero.

Direct evaluation of the stable distribution by inversion of the characteristic function is a complicated procedure, requiring a Fast Fourier transform plus a method for summing the asymptotic expansion. Therefore, very considerable setup calculations are involved, which would have to be repeated for every change in the parameters. It is possible that this effort might be repaid in an extremely large Monte Carlo sample with fixed parameters, but for most applications it is felt the present method of nonlinearly transforming two uniform random variables will be cheaper as well as much simpler to implement.

Paulson *et al.* [15], report success with approximating the inverse distribution using Zolotarev's integral representations for the distribution [19]. Again, in our opinion the present approach will be cheaper as well as much simpler to implement.

An empirical estimate of accuracy was obtained by coding the algorithm a second time, using double precision arithmetic and routines with double precision accuracy for the trigonometric and exponential functions. Suppose $S'(\alpha, \beta', u, w)$ denotes the standard function, as computed in the FORTRAN function RSTAB; let S_D and S_T be the corresponding double precision and true (i.e., exact) values of the function. If

$$|S_D - S_T| \ll |S' - S_T| \ ,$$

then

$$|S' - S_D| \cong |S' - S_T| \ ,$$

so that a comparison of S' and S_D' is a useful test of accuracy. In our case, these estimates may be pessimistic, since the standard algorithm employs some double precision arithmetic and S_D does not increase the precision of these calculations.

For each of $\epsilon(\epsilon = 1 - \alpha)$, β', u and $u'(w = -\ln u')$, we chose five values (2^{-3}, 2^{-2}, 2^{-1}, $1 - 2^{-2}$, $1 - 2^{-3}$) and computed S' and S_D'. The relative differences were typically of the order of 10^{-8}; all calculations were carried out on a Honeywell 6070, with a single precision relative accuracy of 2×10^{-6}. A few of the 625 values were significantly larger, with the four largest between 10^{-6} and 5×10^{-6} (the large values occurred for large ϵ, i.e., small α); these results are available on request.

Some simple timing tests were performed on the same computer. The average execution time for a call to RSTAB was about one millisecond (less for special cases, such as $\beta' = 0$). This can be compared to the typical floating-point operation times on this machine of two to eight microseconds, and also with the typical execution time of about 23 and 73 microseconds for the uniform and exponential generators used here.

Finally, a word on the monetary cost. To generate 1,000 stable pseudorandom variables on this machine at standard (presently available) commercial rates costs about 35 cents.

6. CLOSING REMARK

It is tempting to conjecture that the representations (2.3) and (2.4) must have simple direct interpretations, so that (when properly viewed) the stable character of the variables they define would be immediately evident. However, after some effort, we have been unable to make any such direct interpretation of the representation, even in the Cauchy case.

APPENDIX: FORTRAN PROGRAMS[1]

```
CRSTAB      RANDOM STABLE STANDARDISED FORM
      FUNCTION RSTAB(ALPHA,BPRIME,U,W)
C ARGUMENTS ..
C       ALPHA = CHARACTERISTIC EXPONENT
C      BPRIME = SKEWNESS IN REVISED PARAMETERIZATION
C           U = UNIFORM VARIATE ON (0.,1.), FOR EXAMPLE FROM A UNIFORM
C             PSEUDO-RANDOM NUMBER GENERATOR
C           W = EXPONENTIALLY DISTRIBUTED VARIATE
C
      DOUBLE PRECISION DA,DB
      DATA PIBY2/1.57079633/,PIBY4/.785398163/
      DATA THR1/0.99/
      EPS=1.0-ALPHA
C COMPUTE SOME TANGENTS
      PHIBY2=PIBY2*(U-0.5)
      A=PHIBY2*TAN2(PHIBY2)
      BB=TAN2(EPS*PHIBY2)
      B=EPS*PHIBY2*BB
      IF(EPS.GT.-0.99) TAU=BPRIME/(TAN2(EPS*PIBY2)*PIBY2)
      IF(EPS.LE.-0.99) TAU=BPRIME*PIBY2*EPS*(1.-EPS)*TAN2((1.-EPS)*PIBY2)
C COMPUTE SOME NECESSARY SUBEXPRESSIONS
C IF PHI NEAR PI BY 2, USE DOUBLE PRECISION.
      IF(A.GT.THR1) GO TO 50
C SINGLE PRECISION
      A2=A**2
      A2P=1.+A2
      A2=1.-A2
      B2=B**2
      B2P=1.+B2
      B2=1.-B2
      GO TO 100
C DOUBLE PRECISION
   50 DA=DBLE(A)**2
      DB=DBLE(B)**2
      A2=1.D0-DA
      A2P=1.D0+DA
      B2=1.D0-DB
      B2P=1.D0+DB
C COMPUTE COEFFICIENT
  100 Z=A2P*(B2+2.*PHIBY2*BB*TAU)/(W*A2*B2P)
C COMPUTE THE EXPONENTIAL-TYPE EXPRESSION
      ALOGZ=ALOG(Z)
      D=D2(EPS*ALOGZ/(1.-EPS))*(ALOGZ/(1.-EPS))
C COMPUTE STABLE
      RSTAB=(1.+EPS*D) * 2.*((A-B)*(1.+A*B) - PHIBY2*TAU*BB*(B*A2-2.*A))
     $  /(A2*B2P)
     $ +TAU*D
      RETURN
      END
```

[1] The reader is cautioned that the appendix has not been checked by a referee. Comments by A. D. Schumaker are gratefully acknowledged.

```
CD2           EVALUATE (EXP(X)-1)/X
      FUNCTION D2(Z)
      DOUBLE PRECISION P1,P2,Q1,Q2,Q3,PV,ZZ
      DATA P1,P2,Q1,Q2,Q3/.84006 68525 36483 239 D3
     &,.20001 11415 89964 569 D2
     &,.16801 33705 07926 648 D4
     &,.1801 33704 07390 023 D3
     &,1.D0/
C THE APPROXIMATION 1801 FROM HART ET AL (1968, P. 213)
      IF(ABS(Z).GT.0.1)GO TO 100
      ZZ=Z*Z
      PV=P1+ZZ*P2
      D2=2.D0*PV/(Q1+ZZ*(Q2+ZZ*Q3)-Z*PV)
      RETURN
  100 D2=(EXP(Z)-1.0)/Z
      RETURN
      END


CTAN          TANGENT FUNCTION
      FUNCTION TAN(XARG)
      LOGICAL NEG,INV
      DATA P0,P1,P2,Q0,Q1,Q2
     & /.129221035E+3,-.887662377E+1,.528644456E-1,
     &.164529332E+3,-.451320561E+2,1.0/
C THE APPROXIMATION 4283 FROM HART ET AL(1968, P. 251)
      DATA PIBY4/.785398163/,PIBY2/1.57079633/
      DATA PI/3.14159265/
      NEG=.FALSE.
      INV=.FALSE.
      X=XARG
      NEG=X.LT.0.
      X=ABS(X)
C PERFORM RANGE REDUCTION IF NECESSARY
      IF(X.LE.PIBY4)GO TO 50
      X=AMOD(X,PI)
      IF(X.LE.PIBY2)GO TO 30
      NEG=.NOT.NEG
      X=PI-X
   30 IF(X.LE.PIBY4)GO TO 50
      INV=.TRUE.
      X=PIBY2-X
   50 X=X/PIBY4
C CONVERT TO RANGE OF RATIONAL
      XX=X*X
      TAN=X*(P0+XX*(P1+XX*P2))/(Q0+XX*(Q1+XX*Q2))
      IF(NEG)TAN=-TAN
      IF(INV)TAN=1./TAN
      RETURN
      END


CTAN2         COMPUTE TAN(X)/X
C     FUNCTION DEFINED ONLY FOR ABS(XARG).LE.PI BY 4
C     FOR OTHER ARGUMENTS RETURNS TAN(X)/X, COMPUTED DIRECTLY
      FUNCTION TAN2(XARG)
      DATA P0,P1,P2,Q0,Q1,Q2
     & /.129221035E+3,-.887662377E+1,.528644456E-1,
     &.164529332E+3,-.451320561E+2,1.0/
C THE APPROXIMATION 4283 FROM HART ET AL(1968, P. 251)
      DATA PIBY4/.785398163/,PIBY2/1.57079633/
      DATA PI/3.14159265/
      X=ABS(XARG)
      IF(X.GT.PIBY4)GO TO 200
      X=X/PIBY4
C CONVERT TO RANGE OF RATIONAL
      XX=X*X
      TAN2=(P0+XX*(P1+XX*P2))/(PIBY4*(Q0+XX*(Q1+XX*Q2)))
      RETURN
  200 TAN2=TAN(XARG)/XARG
      RETURN
      END
```

[*Received July 1974. Revised July 1975.*]

REFERENCES

[1] Andrews, D.F., *et al.*, *Robust Estimation of Location: Survey and Advances*, Princeton, N.J.: Princeton University Press, 1972.

[2] Box, G.E.P. and Muller, M.E., "A Note on the Generation of Random Normal Deviates," *Annals of Mathematical Statistics*, 29, No. 2 (1958), 610–11.

[3] Brown, G.W. and Tukey, J.W., "Some Distributions of Sample Means," *Annals of Mathematical Statistics*, 17, No. 1 (1946), 1–12.

[4] Cootner, P., ed., *The Random Character of Stock Market Prices*, Cambridge, Mass.: M.I.T. Press, 1964.

[5] Cross, M.J., "A Catalog of Finite Mean Stable Random Walks," Unpublished manuscript.

[6] DuMouchel, W.H., *Stable Distributions in Statistical Inference*, Unpublished Ph.D. dissertation, Department of Statistics, Yale University, 1971.

[7] Fama, E., "The Behavior of Stock Market Prices," *Journal of Business*, 38, No. 1 (1965), 34–105.

[8] ——— and Roll, R., "Some Properties of Symmetric Stable Distributions," *Journal of the American Statistical Association*, 63, No. 323 (September 1968), 817–36.

[9] Feller, W., *An Introduction to Probability Theory and Its Applications, Vol. 2*, New York: John Wiley & Sons, Inc., 1966.

[10] Gross, A.M., "A Monte Carlo Swindle for Estimators of Location," *Applied Statistics*, 22, No. 3 (1973), 347–53.

[11] Hart, J.F., *et. al.*, *Computer Approximations*, New York: John Wiley & Sons, Inc., 1968.

[12] Ibragimov, I.A. and Chernin, K.E., "On the Unimodality of Stable Laws," *Theory of Probability and Its Applications*, 4, No. 4 (1959), 417–19.

[13] Kanter, M., "Stable Densities Under Change of Scale and Total Variation Inequalities," *Annals of Probability*, 3, No. 4 (1975), 697–707.

[14] ——— and Steiger, W.L., "Regression and Autoregression with Infinite Variance," *Advances in Applied Probability*, 6, No. 4 (1974), 768–83.

[15] Paulson, A.S., Holcomb, E.W. and Leitch, R.A., "The Estimation of the Parameters of the Stable Laws," *Biometrika*, 62, No. 2 (1975), 163–70.

[16] Relles, D.A., "Variance Reduction Techniques for Monte Carlo Sampling from Student Distributions," *Technometrics*, 12, No. 3 (1968), 499–515.

[17] Samuelson, P., "The Mathematics of Speculative Price," *SIAM Review*, 15, No. 1 (1973), 1–41.

[18] Stuck, B.W. and Kleiner, B., "A Statistical Analysis of Telephone Noise," *Bell System Technical Journal*, 53, No. 7 (1974), 1263–312.

[19] Zolotarev, V.M., "On Representation of Stable Laws by Integrals," *Selected Translations in Mathematical Statistics and Probability, Vol. 6*, Providence, R.I.: American Mathematical Society, 1966, 84–88.